TP8

Exercice 1.

```
> restart;
> with(plots):
> with(algcurves):
> F:=X^2*Y^5-X^5*Y^2-2*X*Y^5+X^5+Y^5-Y*X^3-X^2*Y^2-Y^3*X;
```

$$F := -X^5 Y^2 + X^2 Y^5 - 2 X Y^5 + X^5 + Y^5 - X^3 Y - X^2 Y^2 - X Y^3 \quad (1)$$

```
> G:=implicitplot(F,X=-6..5,Y=-6..5,grid=[250,250],color=blue,
  thickness=1):
> display(G);
```




```
> S:=solve({F,diff(F,X),diff(F,Y)},{X,Y});
```

$$S := \{X = 0, Y = 0\} \quad (2)$$

l'origine est le seul point singulier

```
> for m from 0 to infinity do
>     CT:=mtaylor(F,{X=0,Y=0},m+1);
>     if CT<>0 then break end if;
>     end do:
> CT;
```

$$-X^3 Y - X^2 Y^2 - X Y^3 \quad (3)$$

```
> op({allvalues(evala(AFactor(CT)))});
```

$$-X \left(X + \left(-\frac{1}{2} I\sqrt{3} + \frac{1}{2}\right) Y\right) \left(X + \left(\frac{1}{2} I\sqrt{3} + \frac{1}{2}\right) Y\right) Y \quad (4)$$

l'origine est donc un point multiple d'ordre 4 ordinaire

Exercice 2

```
> restart;
> with(LinearAlgebra):
```

```
> M:=<<6,12,18>|<8,12,4>|<4,18,4>|<20,30,10>>;
```

$$M := \begin{bmatrix} 6 & 8 & 4 & 20 \\ 12 & 12 & 18 & 30 \\ 18 & 4 & 4 & 10 \end{bmatrix} \quad (5)$$

```
> U,S,V:=SmithForm(M,output=['U','S','V']);
```

$$U, S, V := \begin{bmatrix} -21 & -4 & 1 \\ -22 & -4 & 1 \\ -660 & -121 & 30 \end{bmatrix}, \begin{bmatrix} 2 & 0 & 0 & 0 \\ 0 & 2 & 0 & 0 \\ 0 & 0 & 174 & 0 \end{bmatrix}, \begin{bmatrix} 3 & 13 & -46 & 0 \\ -4 & -31 & 101 & -5 \\ -1 & 2 & -3 & 0 \\ 1 & 8 & -26 & 2 \end{bmatrix} \quad (6)$$

```
> Equal(S,U.M.V);
```

$$true \quad (7)$$

```
> nl,nc:=Dimension(M);r:=min(nl,nc);
```

$$nl, nc := 3, 4$$

$$r := 3 \quad (8)$$

```
> with(combinat,choose):
```

```
> for k from 1 to r do
>          CL:=choose(nl,k);
>          CC:=choose(nc,k);
>          SM[k]:=[seq(seq(M[a,b],a=CL),b=CC)];
>       end do:
> SM[1];
```

$$\left[\left[\begin{array}{c}6\end{array}\right],\left[\begin{array}{c}12\end{array}\right],\left[\begin{array}{c}18\end{array}\right],\left[\begin{array}{c}8\end{array}\right],\left[\begin{array}{c}12\end{array}\right],\left[\begin{array}{c}4\end{array}\right],\left[\begin{array}{c}4\end{array}\right],\left[\begin{array}{c}18\end{array}\right],\left[\begin{array}{c}4\end{array}\right],\left[\begin{array}{c}20\end{array}\right],\left[\begin{array}{c}30\end{array}\right],\left[\begin{array}{c}10\end{array}\right]\right] \quad (9)$$

```
> SM[2];
```

$$\left[\left[\begin{array}{cc}6 & 8\\12 & 12\end{array}\right],\left[\begin{array}{cc}6 & 8\\18 & 4\end{array}\right],\left[\begin{array}{cc}12 & 12\\18 & 4\end{array}\right],\left[\begin{array}{cc}6 & 4\\12 & 18\end{array}\right],\left[\begin{array}{cc}6 & 4\\18 & 4\end{array}\right],\left[\begin{array}{cc}12 & 18\\18 & 4\end{array}\right],\left[\begin{array}{cc}6 & 20\\12 & 30\end{array}\right],\left[\begin{array}{cc}6 & 20\\18 & 10\end{array}\right],\right. \quad (10)$$
$$\left[\begin{array}{cc}12 & 30\\18 & 10\end{array}\right],\left[\begin{array}{cc}8 & 4\\12 & 18\end{array}\right],\left[\begin{array}{cc}8 & 4\\4 & 4\end{array}\right],\left[\begin{array}{cc}12 & 18\\4 & 4\end{array}\right],\left[\begin{array}{cc}8 & 20\\12 & 30\end{array}\right],\left[\begin{array}{cc}8 & 20\\4 & 10\end{array}\right],\left[\begin{array}{cc}12 & 30\\4 & 10\end{array}\right],$$
$$\left.\left[\begin{array}{cc}4 & 20\\18 & 30\end{array}\right],\left[\begin{array}{cc}4 & 20\\4 & 10\end{array}\right],\left[\begin{array}{cc}18 & 30\\4 & 10\end{array}\right]\right]$$

```
> SM[3];
```

$$\left[\left[\begin{array}{ccc}6 & 8 & 4\\12 & 12 & 18\\18 & 4 & 4\end{array}\right],\left[\begin{array}{ccc}6 & 8 & 20\\12 & 12 & 30\\18 & 4 & 10\end{array}\right],\left[\begin{array}{ccc}6 & 4 & 20\\12 & 18 & 30\\18 & 4 & 10\end{array}\right],\left[\begin{array}{ccc}8 & 4 & 20\\12 & 18 & 30\\4 & 4 & 10\end{array}\right]\right] \quad (11)$$

```
> for k from 1 to r do
>      DM[k]:=map(Determinant,SM[k]);
>              end do:
> DM[1]; # mineurs d'ordre 1
```

$$[6, 12, 18, 8, 12, 4, 4, 18, 4, 20, 30, 10] \quad (12)$$

```
> DM[2]; # mineurs d'ordre 2
```

$$[-24, -120, -168, 60, -48, -276, -60, -300, -420, 96, 16, -24, 0, 0, 0, -240, -40, 60] \quad (13)$$

```
> DM[3]; #mineurs d'ordre 3
```

$$[1392, 0, -3480, 0] \quad (14)$$

```
> for k from 1 to r do
>      d[k]:= igcd(op(DM[k]));
>              end do:
> d[1];d[2];d[3]; #les invariants d□terminantiels
```

$$2$$
$$4$$
$$696 \quad (15)$$

```
> finv:=[d[1],seq(d[k]/d[k-1],k=2..r)]; # les facteurs invariants
```

(16)

$$finv := [2, 2, 174] \quad (16)$$

```
> Uinv:=U^(-1);
```

$$Uinv := \begin{bmatrix} 1 & -1 & 0 \\ 0 & 30 & -1 \\ 22 & 99 & -4 \end{bmatrix} \quad (17)$$

```
> E:=[seq(Column(Uinv,j),j=1..ColumnDimension(Uinv))];
```

$$E := \left[\begin{bmatrix} 1 \\ 0 \\ 22 \end{bmatrix}, \begin{bmatrix} -1 \\ 30 \\ 99 \end{bmatrix}, \begin{bmatrix} 0 \\ -1 \\ -4 \end{bmatrix}\right] \quad (18)$$

```
> U.E[1],U.E[2],U.E[3];;
```

$$\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix} \quad (19)$$

```
> F:=[seq(finv[j]*E[j],j=1..ColumnDimension(Uinv))];
```

$$F := \left[\begin{bmatrix} 2 \\ 0 \\ 44 \end{bmatrix}, \begin{bmatrix} -2 \\ 60 \\ 198 \end{bmatrix}, \begin{bmatrix} 0 \\ -174 \\ -696 \end{bmatrix}\right] \quad (20)$$

```
> Column(M,1);
```

$$\begin{bmatrix} 6 \\ 12 \\ 18 \end{bmatrix} \quad (21)$$

Exercice 3.

```
> restart;
> with(LinearAlgebra):
> A:=<<1,0,1,0,0,1>|<-2,1,-5/2,-1,0,-3/2>|<4,5/2,2,9/2,2,-1/2>|<-2,
  -7/2,-1/2,-7/2,-2,1>|<5,2,5/2,3,3,3/2>|<-4,-5/2,-3,-7/2,-1,1/2>>;
```

$$A := \begin{bmatrix} 1 & -2 & 4 & -2 & 5 & -4 \\ 0 & 1 & \frac{5}{2} & -\frac{7}{2} & 2 & -\frac{5}{2} \\ 1 & -\frac{5}{2} & 2 & -\frac{1}{2} & \frac{5}{2} & -3 \\ 0 & -1 & \frac{9}{2} & -\frac{7}{2} & 3 & -\frac{7}{2} \\ 0 & 0 & 2 & -2 & 3 & -1 \\ 1 & -\frac{3}{2} & -\frac{1}{2} & 1 & \frac{3}{2} & \frac{1}{2} \end{bmatrix} \tag{22}$$

```
> n,n:=Dimension(A);
```

$$n, n := 6, 6 \tag{23}$$

```
> P:=sort(CharacteristicPolynomial(A,X),X);
```

$$P := X^6 - 4X^5 + 16X^3 - 12X^2 - 16X + 16 \tag{24}$$

```
> Q:=sort(MinimalPolynomial(A,X),X); # le polyn^ome minimal est un
  invariant de similitude
```

$$Q := X^5 - 2X^4 - 4X^3 + 8X^2 + 4X - 8 \tag{25}$$

P est le produit des invariants de similitude et Q est le plus grand (pour la relation de divisibilité) or deg (P)=n, deg(Q)=n-1
il ne peux donc y avoir qu'un seul autre invariant de degré 1 qui est nécessairement P/Q

```
> sort(quo(P,Q,X),X); # il n'y a qu'un autre invariant de
  similitude
```

$$X - 2 \tag{26}$$

```
> M:=(-1)^n*CharacteristicMatrix(A,X);
```

$$M := \begin{bmatrix} X-1 & 2 & -4 & 2 & -5 & 4 \\ 0 & X-1 & -\frac{5}{2} & \frac{7}{2} & -2 & \frac{5}{2} \\ -1 & \frac{5}{2} & X-2 & \frac{1}{2} & -\frac{5}{2} & 3 \\ 0 & 1 & -\frac{9}{2} & X+\frac{7}{2} & -3 & \frac{7}{2} \\ 0 & 0 & -2 & 2 & X-3 & 1 \\ -1 & \frac{3}{2} & \frac{1}{2} & -1 & -\frac{3}{2} & X-\frac{1}{2} \end{bmatrix} \tag{27}$$

```
> S,U,V:=SmithForm(M,X,output=['S','U','V']):
> S:=map(sort,S,X);
```

$$S := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & X-2 & 0 \\ 0 & 0 & 0 & 0 & 0 & X^5-2X^4-4X^3+8X^2+4X-8 \end{bmatrix} \quad (28)$$

```
> Equal(S,map(expand,U.M.V));
```

$$true \quad (29)$$

```
> Fi:=remove(type,[seq(S[n-i,n-i],i=0..n-1)],constant); # la liste
  les invariants de similitude
```

$$Fi := [X^5-2X^4-4X^3+8X^2+4X-8, X-2] \quad (30)$$